

### 2.5 MODULACIÓN POR ANCHO DE PULSO SINCRONIZADA GENERACIÓN DE UNA PWM TIPO “UNBUFFERED”

Preparado por: Rangel Alvarado  
Estudiante Graduando de Lic. en Ing. Electromecánica  
Universidad Tecnológica de Panamá  
Panamá, Panamá  
“e-mail”: [issaiass@cwpanama.net](mailto:issaiass@cwpanama.net)  
“web site”: <http://www.geocities.com/issaiass/>

ÍNDICE	
2.5.1 <i>Introducción</i>	235
2.5.2 <i>PWM “Buffered”</i>	236
2.5.3 <i>Registros de Estado y Control en Modo Sincronizado</i>	236
2.5.4 <i>Diagrama de Flujo</i>	238
2.5.5 <i>Código</i>	239
2.5.6 <i>Simulación</i>	244
2.5.7 <i>Conclusión</i>	245
2.5.8 <i>Referencias</i>	246
2.5.9 <i>Problemas Propuestos</i>	247

#### 2.5.1 Introducción

---

La modulación por ancho de pulso tipo “unbuffered” es un método efectivo de control de dispositivos, pero tiene una limitante, el ancho de pulso no puede ser conmutado de un extremo a otro (99% a 1% de ciclo de trabajo). Para situaciones en las que la PWM “unbuffered” no sea adecuada, está la PWM “buffer”; este método es muy parecido al modo de comparación de salida “buffer”.

Para la aplicación se logra que usted:

- Utilice una PWM tipo “buffer” para generar anchos de pulsos que varían entre 1% y 99% de ciclo de trabajo y demostrar la ventaja de unir ambos canales.
- Simulación: utilizar “breakpoints” para visualizar el autosincronismo de los canales.

*Nota: Se asume un dominio de la NT0101, NT0102, NT0103 y se hace referencia cruzada a la misma si es necesario.*

## 2.5.2 PWM “Buffered”





Figura 140. Modo de Funcionamiento de una PWM “Buffer”. (a) Comparación de Salida 1. Inicialmente el control lo posee el registro del Canal 0, hasta que se de el sobreflujo de ltemporizador (TOF). (b) Comparación de Salida 2. Luego del sobreflujo el control es poseído por el registro del Canal 1. El control de la sincronización de canales es controlado por TSC0 el cual alterna entre los valores del canal.

Nota: Ver NT0102, sección 2.2.2, para modo de operación tipo “buffer”.

## 2.5.3 Registros de Estado y Control en Modo Sincronizado

### 2.5.3.1 Registros de Estado y Control para Modo “Buffered” – TSC0

- 0 = 100% de ciclo de trabajo inhabilitado.
- 1 = Si TOV0 es cero, fuerza 100% de ciclo de trabajo.
- 0 = Conmutación al sobreflujo del temporizador inactiva.
- 1 = Conmuta al sobreflujo del temporizador, ver Figura 140(a)



- 0 = Comparación de Salida “Buffered” desactivada.
- 1 = Comparación de Salida “Buffered” activada.
- 1 = Comparación de Salida “Unbuffered”.
- 0 = Captura de Entrada.

Nota: Si ELSX[B:A] ≠ 0:0, la combinación MSXA es válida. CHXMAX tiene efecto en el siguiente período.

Tabla 56. Modo de Comparación de Salida Unbuffered

PWM [Buffered], TOV0 = 1	MS0B	MS0A	ELS0B	ELS0A
Conmuta en la comparación de salida	1	N	0	1
Borra en la comparación de salida	1	N	1	0
Impone uno en la comparación de salida	1	N	1	1

MSXB = Modo de Selección del Canal X, Bit B. [Mode Select Bit B].  
 MSXA = Modo de Selección del Canal X, Bit A. [Mode Select Bit A].  
 ELSXB = Nivel de Selección del Canal X, Bit B. [Edge Level Select Bit B].  
 ELSXA = Nivel de Selección del Canal X, Bit A. [Edge Level Select Bit A].  
 Nota: **No programe la PWM para la primera combinación, puede que genere resultados erróneos al trabajar como PWM, ver Figura 134, problemáticas**

Figura 141 (inferior-izquierda). Registro de Estado y Control del Canal del TIM en PWM. Configure según la tabla 56 para la acción de la PWM. Para mayor detalle de los registros ver NT0101, sección 2.1.3 y NT0102, sección 2.2.2.

**2.5.3.2 Configuración de Comparación de Salida en Modo “Buffered”**

Refiérase en esta sección a la NT0101 – Sección 2.1.3 y NT0102 – Sección 2.2.3.

Configure el Temporizador para una PWM tipo “Buffered” a 60 Hz de frecuencia. Utilice un PS de cero (0), imponga al comparar y habilite las interrupciones del Canal 0. ¿Qué anchos de pulso equivaldrían 1% y 99% de ciclo de trabajo?. Nota: En el modo “buffer”, solo se puede extraer una señal por el Canal 0 de tipo temporizada, el PTD5 sirve ahora como puerto de E/S (entrada/salida).

$$TMOD[H : L] = \frac{t \cdot f_{xtal}}{2^{2+PS}} = \frac{(1/60) \cdot 4.9152 \times 10^6}{2^{2+0}} = 20480_{10} = 5000_{16}$$

$$TCH0[H : L] = D_{1\%} \times TMOD[H : L] = 0.1 \times 5000_{16} = 800_{16} \quad (a)$$

$$TCH1[H : L] = D_{99\%} \times TMOD[H : L] = 0.99 \times 5000_{16} = 4F33_{16}$$




Figura 142. Configuración de una PWM tipo “Buffered”. (a) Valores a depositar en los registros para interactuar con el canal 1 y canal 0 vinculados. (b) Configuración de registro de Estado y Control del canal del temporizador. (c) Registros del Canal con valores de anchos de pulsos variables y período.

## 2.5.4 Diagrama de Flujo

El siguiente programa hace conmutar un LED conectado al puerto PTD4 con una frecuencia de 60 Hz alternándose entre 1% de ciclo de trabajo y 99% de ciclo de trabajo. Realizar la conexión adicional que muestra en la figura lateral derecha a 143(b).




Figura 143. NT0105 – “Timer” – PWM “Buffered”. (a) Programa Principal. Inicializa el temporizador en modo PWM “buffered”. (b) Reinicio del sistema. Al presionar “reset”, el sistema reinicia automáticamente.




Figura 144. NT0105 – “Timer” – PWM “Buffered” – Subrutinas. (a) Inicialización de una PWM. Inicializa el temporizador en modo PWM tipo “buffered” para generar un tren de pulsos en la salida PTD4. (b) Interrupción. Se utiliza la misma interrupción para escribir el canal y activar el “buffer”.

## 2.5.5 Código

```
=====
; ARCHIVO      : NT0105 - Timer - PWM Buffered - 07 07 04.asm
; PROPÓSITO   : Genera una PWM sincronizada de 1% y 99% de ciclo de trabajo.
;              - Observar la configuración del Canal del temporizador.
;              - Generación de una PWM Buffered (variable)
;              P.D.: Agregar una resistencia limitadora al PTD4 de 3.3k
;              con un LED con su cátodo a tierra.
; NOTA        :
;              1 - Observar el resultado de la temporización en el LED PTD4
;              2 - Observar el resultado del registro TIM en la
;                 localidades TSC1, TCH1[H:L], TCH0[H:L].
; REFERENCIA:
;              Advanced Information of MCU68HC908JK1, JK3, JL3...
;              http://www.freescale.com/files/microcontrollers/
;              doc/data_sheet/MC68HC08JL3.pdf
;              Pág. 113 -Función de una PWM tipo Buffer, Sección 10.5.4.2
;              Págs. 114 - 115 - Inicialización de una PWM tipo Buffer,
;              Sección 10.5.4.3.
;              Págs. 117 - 121. - Información de registros de Control
;              del Canal
;              Pág. 124 - Tabla de selección de modos del temporizador,
;              Tabla 10-3.
;
; LENGUAJE    : IN-LINE ASSEMBLER
;
; HISTORIAL
; DD MM AA
; 15 03 03 Creado.
; 06 09 04 Modificado.
=====
;Pasos para iniciar el TIM:
;
; 1 - Definir del registro TSC o registro de
; control del TIM:
;     ¿Necesito o no interrupciones? (TOIE)
;     ¿Cuál que valor de divisor ajustar? (PS[2:0])
; 2 - Definir el registro TMOD[H:L] como módulo máximo de conteo
;     TMOD[H:L] = t*fxtal/(2^(2+PS))
; 3 - Habilitar Interrupciones globales (cli)
; 4 - Habilitar el temporizador (TSTOP = 0)
; 5 - Configurar el vector interrupción ($FFF2)
; 6 - Escribir el código de la interrupción y su retorno (rti)
;
;Pasos para iniciar el Canal del TIM
; 1 - Definir del registro TSCX registro de
; control del canal del TIM:
;     ¿Necesito o no interrupciones? (CHXIE)
;     ¿Input Capture, Output Compare, PWM?
;     ¿Inicio en bajo o en alto? (MSXB,MSXA,ELSXB,ELSXA)
; 2 - Definir el registro TCHX[H:L] como módulo de comparación
```

**NT0105**

**Rev. 1 del 06.08.05**

```

;          TCHX[H:L] = D * TMOD[H:L], D = valor en porcentaje
;          3 - Configurar el vector interrupción ($FFF4 - CH1; $FFF6 - CH0)
;          4 - Escribir el código de la interrupción y su retorno (rti)
;=====

```

```

;=====
;
;          Definiciones del Usuario
;=====

```

```

COPD      equ 0T          ; Bit 0 del registro CONFIG1
BIT5      equ 5T          ; TSC, Bit de Inicio de Conteo, Bit 5 ON
CH0F      equ 7T          ; TSC0, Bit de Salida Comparada
CH1F      equ 7T          ; TSC1, Bit de Salida Comparada
HZ60      equ $5000       ; 60 Hz de frecuencia, TMOD[H:L]
D1%       equ $800        ; 99% de ciclo de trabajo. (TIM is Set on OC)
D99%      equ $4F33       ; 1% de ciclo de trabajo. (TIM is Set on OC)
TRST      equ %00010000   ; TSC, Bit de Reinicio de conteo, Bit 4 ON
TSTOP     equ %00100000   ; TSC, Bit de pausa, Bit 5 ON
CH0IE     equ %01000000   ; Habilito interrupciones, canal 1
MS0B      equ %00100000   ; Output Compare Tipo Buffered activado
MS0A      equ %00010000   ; Output Compare Tipo Unbuffered
ELS0B     equ %00001000   ; Bit de Selección de flanco
ELS0A     equ %00000100   ; Bit de Selección de flanco, activa...
; la señal en uno lógico al comparar
TOV0      equ %00000010   ; TSC0, Bit de Cambio de estado al
; sobreflujo del temporizador

```

```

;=====
;
;          Mapa de Memoria del Microcontrolador
;=====

```

```

;
;          Registro de Configuraciones
;=====

```

```

CONFIG1   equ $001F       ; Vectores de configuración
;=====

```

```

;
;          Registro de Temporizador
;=====

```

```

TSC       equ $0020       ; Dirección, registro de estado y control del
; TIM
TCNTH     equ $0021       ; TCNT, Registro almacenador de cuenta del
; módulo, registro alto.
TCNTL     equ $0022       ; TCNT, Registro almacenador de cuenta del
; módulo, registro bajo.
TMODH     equ $0023       ; TMODH, Registro de cuenta del
; módulo, registro alto.
TMODL     equ $0024       ; TMODH, Registro de cuenta del módulo,
; registro bajo.
TSC0      equ $0025       ; Dirección, registro de estado y control del
; TIM Canal 0
TCH0H     equ $0026       ; TCH0H, Registro almacenador de
; comparación de la salida, registro alto.
TCH0L     equ $0027       ; TCH0L, Registro almacenador de
; comparación de la salida, registro bajo.
TSC1      equ $0028       ; Dirección, registro de estado y control del

```

```

; TIM Canal 1
TCH1H      equ $0029      ; TCH1H, Registro almacenador de
; comparación de la salida, registro alto.
TCH1L      equ $002A      ; TCH1L, Registro almacenador de
; comparación de la salida, registro bajo.

;=====
;
;                      Memoria FLASH
;=====
FLASH_START equ $EC00      ; Puntero - Mem.FLASH

;=====
;
;          Vectores de Usuario
;=====
TIMCH0H    equ $FFF6      ; Vector de Sobreflujo del TIM, Canal 0(Alto)
RESET_VEC  equ $FFFE      ; Puntero del RESET

;=====
; OBJETIVO   : Inicio de Codif. del Ensam-
;              blador en Memoria FLASH.
;=====
org FLASH_START      ; Inicio Mem. FLASH

;=====
; OBJETIVO   : Configura el TIM para generar
;              una PWM de 1% y 99% de ciclo
;              de trabajo.
;=====
START
    rsp                ; Inic.Stack = $00ff
    bset COPD,CONFIG1 ; Desactiva watchdog
    clra               ; Borra A
    jsr INIT_TIM       ; Inicializa TIM
    cli                ; Habilita Interrupciones
    bclr BIT5,TSC      ; Inicia el temporizado (TSTOP = 0)
ESPERA
    wait               ; Espera la interrupción.
    bra ESPERA         ; Salta al modo de bajo consumo
    
```

```

=====
; INIT_TIM      : Inicializa el TIM
; OBJETIVO     : Inicialización del temporizador.
;              : PWM de tipo Buffered
; ENTRADA      : Ninguna
; SALIDA       : Ninguna
; REGISTROS    :
; AFECTADOS   : TSC, TCH0H, TCH0L, TSC0, H:X
=====
INIT_TIM
    ldhx #D1%                ; 1% de ciclo de trabajo
    sthx TCH0H               ; Almacena en TCH0
    ldhx #D99%              ; 99% de ciclo de trabajo
    sthx TCH1H               ; Almacena en TCH1
    ldhx #HZ60               ; 60 Hz de frecuencia
    sthx TMODH               ; Almacena en TMOD
    mov #{TRST|TSTOP},TSC    ; Reinicio el temporizador
                                ; PS[2:0] = 0, el temporizador se encuentra
                                ; detenido por defecto.
    mov #{CH0IE|MS0B|MS0A|ELS0B|ELS0A|TOV0},TSC0
                                ; Habilito Interrupciones del canal y configuro
                                ; una PWM Buffered, conmuto al sobre
                                ; flujo del temporizador el puerto.
    rts                      ; retorna

=====
; TIMCH0L      : Interrupción del Canal 0 del
;              : TIM y sobreflujos simulada
; OBJETIVO     : Reconocer las interrupciones
;              : Cambia al otro tipo de pulso
;              : en buffer.
; ENTRADA      : Ninguna
; SALIDA       : Ninguna
; REGISTROS    :
; AFECTADOS   : TSC0, TSC1, TCH0H,
;              : TCH0L, TCH1H, TCH1L,
;              : H:X
=====
TIMCH0L
    bclr CH0F,TSC0           ; Reconoce la interrupción del Canal 0
    brset CH1F,TSC1,CH1     ; Si se da la interrupción del Canal 1,
                                ; cambia Canal 0
    ldhx #D99%              ; Sino, carga 99%
    sthx TCH1H               ; Almacena en Canal 1
    bra OUTTIM              ; Salir
CH1    bclr CH1F,TSC1       ; Reconoce la interrupción del Canal 1
    ldhx #D1%                ; Carga 1%
    sthx TCH0H               ; Almacena en Canal 0
OUTTIM
    rti                      ; Retorna de la interrupción

```



```
=====
; OBJETIVO : Inicializa el Vector de Reset
;           ; Arranque del programa en la
;           ; memoria Flash y búsqueda de
;           ; la interrupción del canal del
;           ; temporizado y de sobreflujo.
;=====
;===== Vectores del TIM, sobreflujo y comparación =====
org TIMCH0H ; Inicio de los vectores del TIM
dw TIMCH0L ; Desborde del contador del Canal
;===== Vector de Reinicio de Sistema =====
org RESET_VEC ; Puntero Vec - RESET
dw START ; al darse reset salta a Start
```

*Listado 14. NT0105 – "Timer" – PWM "Buffered". El programa responde a inicializar el temporizador, canal 0 para generar una PWM "Buffered", con una frecuencia de 60 Hz y alternancias de los ciclos de trabajo del 1% y 99% ciclicamente.*

## 2.5.6 Simulación

Para el uso de esta sección se asume tener dominio de la NT0006 (Sección 1.6.4), NT0009 (Sección 1.9.5), NT0102 y NT0104 (Sección 2.4.3).




Figura 145. Simulación del Programa Principal. (a) “Breakpoint”. Sitúe varios “breakpoints” para observar el modo de operación tipo “buffer” de ambos canales unidos. (b) “Buffer” Inicial, Canal 0. Luego de la primera interrupción, el canal 0, comandado por TCH0 tenía el control. (c) Siguiendo valor del “buffer”, Canal 1. Luego del sobreflujo del temporizador (TOF = 1), el canal 1 entra a interactuar. La operación se repite alternando los canales 0 y 1 sucesivamente.

- (a) Inicie WinIDE.
- (b) Cargue el archivo NT0105 – Timer – PWM Buffered – 07 07 04.asm.
- (c) Compile.
- (d) Entre al simulador.
- (e) Añada “breakpoints” según la figura 146(a).
- (f) Corra su programa paso a paso y en la sección de llamado a la rutina de inicialización del temporizador (INIT\_TIM) observe la configuración de una PWM tipo “buffer”.
- (g) Corra su programa y espere a la llegada del “breakpoint”. Cuando la cuenta iguala al registro del canal, se da una interrupción comandada por el “buffer” inicial TCH0, figura 146(b).
- (h) Nuevamente corra su programa y observe que cuando existe el sobreflujo del temporizador (TOF = 1) y se halla desbordado la cuenta del canal 1 (CH1F = 1), el canal 1 todavía tenía control del temporizado, figura 146(c).
- (j) Si desea “quemar” su pastilla, revisar la NT0009, Sección 1.9.5.

Nota: Recuerde mover su compilación condicional si desea que su programa corra en tiempo real.

```
    ; $SET ICS08  
    $SETNOT ICS08
```

## 2.5.7 Conclusión

---

La característica principal del modo PWM de tipo “buffer” permite eliminar el problema de sincronismo y evita mal funcionamiento de la señal de pulso a modular.

Para fin demostrativo se realizó una PWM tipo “buffer” que cambia de 1% a 99% de ciclo de trabajo y mediante simulación se pudo visualizar el sincronismo que existe entre ambos canales del temporizador.

## 2.5.8 Referencias

---

### 2.5.8.1 Información Avanzada sobre el Microcontrolador

(a) [http://www.freescale.com/files/microcontrollers/doc/data\\_sheet/MC68HC08JL3.pdf](http://www.freescale.com/files/microcontrollers/doc/data_sheet/MC68HC08JL3.pdf)

*Pág. 113 – Función de una PWM tipo “Buffer”, Sección 10.5.4.2*

*Págs. 114 – 115 – Inicialización de una PWM tipo “Buffer”, Sección 10.5.4.3.*

*Págs. 117 – 121. - Información de registros de Control del Canal*

*Pág. 124 – Tabla de selección de modos del temporizador, Tabla 10-3.*

### 2.5.8.2 Manual de Referencia del CPU

(a) [http://www.freescale.com/files/microcontrollers/doc/ref\\_manual/CPU08RM.pdf](http://www.freescale.com/files/microcontrollers/doc/ref_manual/CPU08RM.pdf)

### 2.5.8.3 Página “web” sobre esta Nota Técnica

(a) <http://www.geocities.com/issaiass/>

## 2.5.9 Problemas Propuestos

---

2.5.9.1 Realizar una PWM de tipo “buffer” de 10kHz de frecuencia, iniciando en 10% de ciclo de trabajo y variando hasta 80% de ciclo. Utilice un preescalar de cuatro (PS = 4), habilite las interrupciones necesarias.

2.5.9.1 Realizar una PWM de tipo “buffer” de 1 Hz de frecuencia, de incremento de 10% de ciclo de trabajo para el Canal 0 y 20% de ciclo de trabajo para el Canal 1.